
プログラミング言語論第1回
 変数と関数の定義と評価のモデル

情報工学科 山本修身

この講義について
プログラミング言語のしくみについて学ぶ．実世界で使われているプ
ログラミング言語には多様なものがあるが，色々なプログラミング言
語に共通するしくみが存在する．それらのしくみをどのような言語で
説明しても構わないが，多くの仕組みを包含するような言語の方が説
明しやすい．C言語は計算機上で高速にプログラムを実行させるのに
は向いているが，このようなプログラム言語のしくみの多くを持たな
いという点において，このような作業に向いていない．

ここでは，Lisp言語の方言であるScheme言語を採用し，プログラミ
ング言語のしくみについて詳しくみていく．

講義データ
講義名： プログラミング言語論
担当： 山本修身 (osami@meijo-u.ac.jp)
対象年次： 3年次
教室： N-206
日時： 金曜日3限 13:10～14:40
ホームページ： http://osami.s280.xrea.com/ProgLang2015/
授業のスタイル： 講義（教室で説明をします．授業内容について講義
のあと各自確認をすること．特に解説したプログラムについては各自
確認すること．）

講義の目標： プログラミング言語の色々なしくみについて理解する．
そのために，単純なしくみをもつプログラミング言語 Schemeを学
び，そこから色々なしくみを探っていく．

プログラムの実行環境： Kawa (Scheme) (1)
この講義で扱う言語は，Schemeである．Scheme言語は色々な実現
があるが，ここでは，Java言語で書かれたKawaを用いて説明する．
Javaが動く環境であれば，動かすことができる．インタプリターなの
で，普通に立ち上げるとプロンプトがでてくる．

osami-2:yama502> java -jar kawa-1.7.jar 
#|kawa:1|# (car '(a b c))
a
#|kawa:2|# (* 123123132 43234234243)
5323134329619809076
#|kawa:3|# 

http://www.gnu.org/software/kawa/kawaのホームページ：

プロンプト

プロンプトを表示されて，それに対して人間が入力し，さらにそれに
対してインタプリターが結果を返すというしくみをREPL (read-
eval-print loop) と呼ぶ．

プログラムの実行環境： Kawa (Scheme) (2)
プログラムファイルを実行させる場合には，以下のようにする．

(define (exp a n)
  (if (zero? n) 1
      (let ((m2 (exp a (quotient n 2))))
        (if (even? n) (* m2 m2)
            (* m2 m2 a)))))

(display (exp 2 1000))
(newline)

プログラム： sample.scm

osami-2:yama509> java -jar ./kawa-1.7.jar sample.scm
107150860718626732094842504906000181056140481170553360744375038837035
105112493612249319837881569585812759467291755314682518714528569231404
359845775746985748039345677748242309854210746050623711418779541821530
464749835819412673987675591655439460770629145711964776865421676604298
31652624386837205668069376
osami-2:yama510> 

MS Windows 上での kawaの実行
Kawaはjarファイルとして配られているので，OSなどによりファイルを変える必要はな
い．基本的には，前述の例と同様に，java -jar ./kawa-1.7.jar sample.scm のように
実行する．コ
マンドプロン
プトで実行し
ても良いが，
たとえば，
ActiveState
社のkomodo
エディタを用
いれば以下の
ようになる．

色々なScheme言語処理系
Scheme言語の処理系にはコンパイラおよびインタプリタがある．本
講義では主にインタプリタを用いて説明するが，効率が問題になる場
合にはコンパイルして実行形式に変換して動かすことができる．

いくつかのインタプリタはweb上で動作する．たとえば，codepadで
はMzSchemeが動作するし，ideoneではguileが動作する． また，
repl.it 上では biwascheme が利用できる．作ったプログラムの動作
を確認するだけであれば，このようなweb上のシステムを利用するこ
とができる．

ある程度効率よく動かすには，schemeシステムをインストールする
必要があり，前述のkawa以外にも mit-scheme，gambit scheme, 
gauche など色々な処理系が開発されている．

プログラムとデータの構成するS-式 (1)
SchemeのプログラムはS-式 (S-expression) と呼ばれるデータで構
成されている S-式は以下のようなものである．Schemeで現れるデー
タはS-式以外にない．

1.アルファベット，数字，_, -, ?, <, >, +, *, / などで構成される
アトムはS-式である．

2.数はS-式である．
3.文字列rはS-式である．
4. s1, ..., snがS-式のとき，(s1 s2 … sn)はS-式である．
5. s1, ..., sn, sn+1 が S-式のとき，(s1 s2 … sn . sn+1) はS-式である．

Schemeの文法にはこの他に ’ (クオー
ト）や , （コンマ）などの マクロの記法
があるがここでは説明に含めない．

(a b c d) のように()で構成さ
れたものを リスト (list) と
呼ぶ．

プログラムとデータの構成するS-式 (2)

(abc def 123 443 8273)

(+ 232 23312)

(find-current-list this-list foo bar)

((2 3 4) (4 3 2) (6 5 4) (1 1 1))

(((((34 45)))))

(a b c d e f g)

(2343 “abcde” abcde 3we . 2343)

((abc . def) (mmm . 332) (mmm . 232))

S-式の例

評価できる S-式
S-式によって色々な構造をもつデータを表現することができる．
SchemeではプログラムもS-式によって表現する．

さらに，Schemeでは表現されたプログラムを評価 (evaluate) する
ことにより結果が得られる．評価することが計算そのものである．評
価できる式は以下のような一定の構造をもつ必要がある．

(関数 引数1 引数２ ... )
ここで，「関数」は入力を出力に変える能力をもつものである．「引
数1」, 「引数２」, ... はこの関数に与えられるデータである．この関
数の出力はこの式の評価値となる．

関数引数1 引数２ ... 式の値

S-式の評価の例 (1)

(+ 1 2 3 4 5) => 15

(* 1 2 3 4 5) => 120

(/ 1 2 3 4 5) => 1/120

#|kawa:5|# (+ 1 2 3 4 5)
15
#|kawa:6|# (* 1 2 3 4 5)
120
#|kawa:7|# (/ 1 2 3 4 5)
1/120
#|kawa:8|# 

+1 2 3 4 5 15

*1 2 3 4 5 120

/1 2 3 4 5 1/120

S-式の評価の例 (2)
もっと複雑な式を作ることもできる．

1
1 · 2

+
1

2 · 3
+

1
3 · 4

+
1

4 · 5
+

1
5 · 6

(+ (/ 1 1 2) (/ 1 2 3) (/ 1 3 4) (/ 1 4 5) (/ 1 5 6))

=> 5/6

この場合，まず内部のS-式を評価して値を確定させてからそれを用い
て外側のS-式を評価する．

#|kawa:10|# (+ (/ 1 1 2) (/ 1 2 3) (/ 1 3 4) (/ 1 4 5) (/ 1 5 6))
5/6

S-式の評価の例 (3)
文字列の連結： “abc” + “def” + “xyz”

#|kawa:14|# (string-append "abc" "def" "xyz")
abcdefxyz

文字列の長さ： |”meijo-university-information-engineering”|
#|kawa:16|# (string-length "meijo-university-information-engineering")
40

文字列を大文字にする：”hello” => “HELLO”
#|kawa:17|# (string-upcase "hello")
HELLO

#|kawa:18|# (string-upcase "Meijo University")
MEIJO UNIVERSITY

変数の導入
Schemeでは変数を定義して利用することができる．変数を定義する
方法はいくつかあるが，一番単純な方法は define を用いることであ
る．Schemeでは変数は型をもたない（データは型を持つ）．した
がって，任意の変数に任意のデータ（S-式）を結ぶことができる．変
数を定義する場合には以下のようなS-式を評価する．

(define 変数名 値）
#|kawa:19|# (define r 2.0)
#|kawa:20|# (define pi 3.1415926)
#|kawa:21|# (define s (* pi r r))
#|kawa:22|# s
12.5663704

S-式の表示
S式を画面に表示するには display 関数を用いる．displayは特殊な出
力（void) を返す．画面に表示されることが，この関数については重
要である．関数 display に渡される引数はもちろん評価されてから表
示される．

(display S-式)
display関数は改行を出力しない．そのため改行を出力するには
newline関数を用いる．newline関数は引数を持たない．

(newline)

簡単なプログラム例 (1)

円の面積を計算するプロ
グラムは以下のように書
ける．プログラムは評価
すべき式を順にソース
ファイルに並べていけば
良い．改行は関係ない
が，行の中に ; があると
それ以降，改行までがコ
メント文となる．

;; プログラム例                                                                 

(define r 2.0)        ; 半径                                                    
(define pi 3.1415926) ; 円周率                                                  
(define s (* pi r r)) ; 面積の計算                                              
(define el (* 2 pi r)); 円周長の計算                                            

;; 計算結果表示                                                                 

(display "r = ")(display r)(newline)
(display "s = ")(display s)(newline)
(display "el = ")(display el)(newline)

simpleProg.scm

OMacBook:yama507> java -jar kawa-1.7.jar simpleProg.scm
r = 2.0
s = 12.5663704
el = 12.5663704
OMacBook:yama508> 

プログラムの中で漢字を使う場合
には，UTF-8で記述すること．

変数への値の代入
set! を用いて，すでに定義されている変数の値を変更することができ
る．（変数が定義されていないと，エラーになるのが本来だが，
kawaの場合には適当に変数が定義されるようである．）

(set! 変数名 値）

#|kawa:8|# (define a 2)
#|kawa:9|# (set! a (* a a a a))  ; 2^4
#|kawa:10|# (set! a (* a a a a)) ; 2^16
#|kawa:11|# (set! a (* a a a a)) ; 2^64
#|kawa:12|# (set! a (* a a a a)) ; 2^256
#|kawa:13|# a
11579208923731619542357098500868790785326998466564
0564039457584007913129639936

簡単なプログラム例 (2)
指数関数の底 e を計算する．exは以下のように近似することができる．

ex = 1 + x +
x2

2!
+

x3

3!
+ · · · +

xn

n!

n = 20とすれば，以下のようなプログラムが得られる．

=
��

· · ·
��x

n
+ 1

� x

n� 1
+ 1

�
· · ·

�
x

2
+ 1

�
x + 1

(define p (+ 1.0 (/ 1.0 20)))
(set! p (+ (/ p 19) 1))
(set! p (+ (/ p 18) 1))
(set! p (+ (/ p 17) 1))
(set! p (+ (/ p 16) 1))
(set! p (+ (/ p 15) 1))
(set! p (+ (/ p 14) 1))
(set! p (+ (/ p 13) 1))
(set! p (+ (/ p 12) 1))
(set! p (+ (/ p 11) 1))

(set! p (+ (/ p 10) 1))
(set! p (+ (/ p 9) 1))
(set! p (+ (/ p 8) 1))
(set! p (+ (/ p 7) 1))
(set! p (+ (/ p 6) 1))
(set! p (+ (/ p 5) 1))
(set! p (+ (/ p 4) 1))
(set! p (+ (/ p 3) 1))
(set! p (+ (/ p 2) 1))
(set! p (+ (/ p 1) 1))
(display p)
(newline) #|kawa:27|# (load "exp.scm")

2.718281828459045

関数を定義する
いくつかのまとまった動きを一つをまとめて一つの命令にすることが
できる．まとまった動きは関数として呼び出して実行させることがで
きる．関数定義は変数定義の同じように以下のように行う．

(define (関数名 引数1 ... 引数n) 式1 式2 ... 式m）

関数を呼び出すと，式1 式2, ... 式mが順に評価されて，式の評価値は
式mの評価値となる．

#|kawa:28|# (define (dd x) (+ x x))
#|kawa:29|# (dd 23)
46
#|kawa:30|# (dd 3847)
7694

関数本体

簡単なプログラム例 (2)：改良
前述のプログラム例は関数を用いることでより見通しが良くなる．

(define p (+ 1.0 (/ 1.0 20)))
(define (iter n) (set! p (+ (/ p n) 1)))
(iter 19)(iter 18)(iter 17)(iter 16)(iter 15)
(iter 14)(iter 13)(iter 12)(iter 11)(iter 10)
(iter 9)(iter 8)(iter 7)(iter 6)(iter 5)
(iter 4)(iter 3)(iter 2)(iter 1)
(display p)
(newline)

関数iterの内容はいちいち書かなくてもiterを用いて表現することがで
きる．これにより，プログラムの行数は飛躍的に少なくなる．しか
し，「繰り返し」ができないので，ひとつずつ命令を書かなければな
らないの不自由である．

条件分岐の導入 (1)
ある条件が成り立つか否かによって，評価値を変えるような動作を導
入することにより，プログラムをより柔軟に動かすことが可能にな
る．そのための命令として以下のような if を導入する．

(if 条件式 式1 式2）

(if 条件式 式1）
ifには2つの形式がある．最初の形式は条件式が真であれば式1を評
価，偽であれば，式2を評価して式の値とする．後者は条件式が真で
あれば，式1を評価して値とし，偽であれば値が無いことを意味する
void値を返す．

#|kawa:1|# (define n 23)
#|kawa:2|# (if (< n 30) (display "yes"))
yes
#|kawa:3|# (if (>= n 30) (display "yes"))
#|kawa:4|# (if (odd? n) (display "yes"))
yes
#|kawa:5|# 

条件分岐の導入 (2)
さらに別のタイプの条件分岐を導入する．ある条件が成り立つとき，
一連の式を評価する命令である．

(when 条件式 式1 式2 ... 式n）

条件式が真のとき，式1, 式2, ... , 式n を順に評価し，式の値を式nの
値とする．

簡単なプログラム例 (2)：改良の改良
関数 work を以下のように定義する．(work n) は最初に (iter n)を呼
び出して，その後，(work (- n 1)) を呼び出して一つ小さなnについて
実行する．ただし，nが0になったらやらない．これで繰返しが表現で
きる．

(define p (+ 1.0 (/ 1.0 20)))
(define (iter n) (set! p (+ (/ p n) 1)))
(define (work n)
  (when (> n 0) (iter n) (work (- n 1))))
(work 19)
(display p)
(newline)

このプログラムの問題点はpという変数にいちいち代入し直している
点で，繰返しをいちいち手で表現しなくて済んでいるので，そこは良
いが，今ひとつ美しくない．

簡単なプログラム例 (2)：改良の改良の改良
そこで，全体的にpの内容を関数の引数として持ち回るプログラムを
作る．この場合，workの引数pに値がどんどん更新されて入って行
く．最初の値は計算して与える．

(define (work n p)
  (if  (= n 0)
       p
       (work (- n 1) (+ (/ p n) 1))))
(display (work 19 (+ 1.0 (/ 1.0 20))))
(newline)

以前のプログラムに比べるときれいな形になったが，割り算をしてか
ら1を足すという部分が2カ所出て来たり，まだ今ひとつすっきりしな
い．

簡単なプログラム例 (2)：改良の改良の改良の改良
そこで，n = 20の部分も分離しないで，同じプログラムで表現するこ
とにして以下のようなプログラムが得られる．このプログラムは簡潔
で美しい形になっている．

(define (work n p)
  (if  (= n 0)
       p
       (work (- n 1) (+ (/ p n) 1))))
(display (work 20 1.0))
(newline)

Schemeではfor文やwhile文がなくてもプログ
ラムの繰返しを書くことができる（C言語でも同
様の呼び出しによって書くことはできるが，ス
タックを消費するので現実的ではない）．

2分法で平方根を求める (1)
2分法とは，区間 [l, r]を半分に狭めることによって，方程式の根
を求める方法である．JavaScriptでのプログラムは以下のようなにな
る．ここではwhile文を用いて，繰り返し計算が行われている．

function binary(){
   var EPS = 1.0e-6
   var left = 0.0
   var right = 2.0
   while (right - left > EPS){
      var m = (left + right) / 2
      var v = m * m - 2
      if (v < 0) left = m
      else right = m
   }
   puts(left + ", " + right)
}

binary()

1.4142131805419922, 1.4142141342163086

0 0.5 1 1.5 2 2.5 3 3.5 4 4.5

-2

-1

1

2

3

4

5

6

left = 0.0 right = 2.0

left = 1.0 right = 2.0

m = 1.0

f(m) = -1.0

2分法で平方根を求める (2)
繰り返しを行うのではなくて，binaryの中からbinaryを呼び出すこと
で繰り返しの動作を表現する．

(define (binary left right f eps)
  (define m (/ (+ left right) 2))
  (if (< (- right left) eps) left
      (if (< (f m) 0)
          (binary m right f eps)
          (binary left m f eps))))

(define (f x) (- (* x x) 2))
(display (binary 0.0 2.0 f 1.0e-6))
(newline)

function binary(){
   var EPS = 1.0e-6
   var left = 0.0
   var right = 2.0
   while (right - left > EPS){
      var m = (left + right) / 2
      var v = m * m - 2
      if (v < 0) left = m
      else right = m
   }
   puts(left + ", " + right)
}

binary()

OMacBook:yama997> ../kawa prog1.scm
1.4142131805419922

関数評価に関するルール
関数を評価する場合のルールについて以下にまとめる．

(func1 a1 a2 ... an)

1. S式のなかのそれぞれの部分式を評価する．部分式が変数で
あれば，それが結び付けられている値を取り出し，S式であ
れば，それをさらに評価する．また，数であればその数その
もの，文字列であれば，その文字列とする．

2. 最初に置かれている値（func1に対応）を関数の実体だと考
え，それに続く値（a1, a2,..., anに対応）を引数の値だと考
えて，関数を実行しその値をこの式の値とする．

関数評価の例（１）
単純な式の場合，それぞれのの記号に結び付けられている実体を呼び
出して，関数評価が行われる．

(* (+ 1 2)    (+ 2 3)    (+ 3 4)     (+ 5 6))

(*の値   3         5           7          11)

(+の評価 1 2)

3

(+の評価 2 3)

5

(+の評価 3 4) (+の評価 5 6)

7 11

1155

掛け算の実行

足し算
の実行

足し算
の実行

足し算
の実行

足し算
の実行

if や defineは関数ではない
以前定義した if や define は評価の仕方が関数と異なる．すなわち，
これらは関数ではない．これらは特殊形式 (special form) と呼ぶ．

(if (< a b)
    (+ a b)
    (- a b))

上の例の場合，(< a b)がtrue (#t) であれば(+ a b)を計算し，
false (#f) であれば，(- a b)を計算するが，もし if が関数であれ
ば，全体を計算する前に両方共計算されていることになる．以下のプ
ログラムを実行すると foobar と出力される理由を考えよ．

(define (my-if cond expr1 expr2)
  (if cond expr1 expr2))

(my-if (= 10 2) (display "foo") (display "bar"))
(newline)

(define (foo x)
    (* x 2))

関数評価の例（２）
以下のようにいくつかの関数を定義する．関数workの式が評価される
過程をみてみる．
(define pi 3.1415926)
(define (circumference r) (* 2 pi r))
(define (square r) (* pi r r))

(define (work r)
  (display (circumference r))
  (newline)
  (display (square r))
  (newline))

(work 5.0)

(work 5.0)

(workの値 5.0)

r = 5.0

(display (circumference r))

(circumferenceの値 5.0)

環境

r = 5.0 環境

 (* 2 pi r)

 (*の値 2 3.1415926 5.0)

31.415926

(newline)

(newlineの値)

値なし

．．．
(displayの値 31.415926)

関数の実行 関数の実行
値なし

関数評価の例（３）
前述のbinary関数の場合，関数定義され
ているので，多少事情が異なる．

(define (binary left right f eps)
  (define m (/ (+ left right) 2))
  (if (< (- right left) eps) left
      (if (< (f m) 0)
          (binary m right f eps)
          (binary left m f eps))))

(binary 0.0 2.0 f 1.0e-6)

(binaryの値 0.0 2.0 fの値 1.0e-6)

left = 0.0,  right = 2.0,  f = fの値,  eps = 1.0e-6

m = 1.0, (f m) = (fの値, 1.0)=-1.0

(define (f x) (- (* x x) 2))

m = 1.0, (f m)

(fの値, 1.0)

x = 1.0

(- (* x x) 2)

(-の値 (* x x)の評価 2)

(*の値 1.0 1.0) = 1.0

(-の値 1.0 2)

-1.0

(binaryの値 1.0 2.0 fの値 1.0e-6)
(binary m right f eps)

環境

環境

環境

関数の実行

関数の実行

今回のまとめ (1)
S式によるデータ（プログラム）の表現

(1 2 3 (5 4 3 . 8) 2)

(abc (1 2) . (def zef))

数の四則演算
(+ 1 2 3 4 5) = 1 + 2 + 3 + 4 + 5 = 15

(- 20 3 2 1) = 20 - 3 - 2 - 1 = 14

(* 2 3 4 5 6) = 2 * 3 * 4 * 5 * 6 = 720

(/ 2 3 4) = 2 / 3 / 4 = 1/6

(+ (* 2 3) (* 4 5)) = 2 * 3 + 4 * 5 = 26

Schemeでは出来る限り正確な演算を行
う．多くの場合，有理数の割り算などでは
分数をデータとして扱うことができる．

今回のまとめ (2)

関数の定義
(define (sqr x) (* x x))

(define (関数名 引数1 引数2 ... 引数n) 式1 式2 ... 式m)
本体

データの表示

(display “abc”)

(display (+ 1 2 3))

(newline)

変数の定義
(define a 20)

(define b “abcde”)

今回のまとめ (3)
定義されている変数の内容を変える

(set! a 34)

(set! b “meijo”)

成り立つ条件によって式の値を変える

(if (< a 0) (- a) a)

(if 条件式 式1 式2)

(when (< m 1) (set! m (+ 1 2)) (* m m))

(when 条件式 式1 式2 .. 式n) whenは条件が成り立たないとき何も値を
返さない．

